

```

DCB = "13";

side = "T";

len = 197; (* Length of the interface *)

wid = 25.4; (* Width of the DCB arm *)

fact = 20; (* Scaling factor for the images *)

img = Import[NotebookDirectory[] <> "DCB-" <> DCB <> "-original.jpg"];
(* Import the image *)

img = ImageRotate[img]; (* The image is rotated *)

(* The rotated image of the specimen will appear. Right-
click on the image and select 'Get Coordinates' to select the interface area either on
the top or the bottom arm. This is done by clicking the angles of the interface are,
which should be done in the following order: bottom-left, top-left, bottom right,
top-right. The last click (top-right) should be right one and then from the
menu 'Copy Graphics Selection' should be selected. The selection should
be then pasted in the square brackets of the following line of code,
i.e. inside the 'To Expression' function. *)

corners = ToExpression[
  {{877.3589888700251, 559.4948122995659}, {866.9760422561785, 985.1956234672701},
  {4044.157706093193, 624.3882286361062}, {4041.561969439731, 1003.3657800415014`}}];
newCorners = {{0, ImageDimensions[img][[2]]}, {0, ImageDimensions[img][[2]] - fact * wid},
  {fact * len, ImageDimensions[img][[2]]},
  {fact * len, ImageDimensions[img][[2]] - fact * wid}};
t = Last@FindGeometricTransform[newCorners, corners];
imgT = ImageCrop[ImagePerspectiveTransformation[img, t, DataRange -> Full],
  fact {len, wid}, {Right, Bottom}];

(* The image is adjusted manually using two parameters,
namely contrast and brightness, in the 'ImageAdjust' function. *)

imgTC = ImageAdjust[ColorSeparate[imgT, "RGB"][[3]], {3.5, -0.2}];

(* The image is then binarised, i.e. transformed to black and white. *)

P1 = Binarize[imgTC];

P1 = ImageResize[P1, 1973]; (* the image is resized *)
Export[NotebookDirectory[] <> "DCB-" <> DCB <> "-" <> side <> "-2D.csv", ImageData[P1]]
(* The image is exported to a 254x1973 matrix consisting of zeros and
ones representing black and white squares of dimensions 0.1x0.1 mm. *)

```